

Paper 191-25

OS/390 SAS® Version 8 WEB GIFS and HTML Using ODS & SAS/GRAPH® Software and MXG®

Neal Musitano Jr.
Department of Veterans Affairs
Philadelphia Benefits Delivery Center
Philadelphia, Pennsylvania

ABSTRACT

OS/390® mainframe computer performance data is collected using IBM System Management Facilities (SMF) and Resource Measurement Facility (RMF) data. The raw SMF/RMF data is processed with MXG® and SAS® software into an easily usable SAS® format.

This paper utilizes that performance data. The MXG® Performance Database, i.e., daily, weekly or monthly contains millions of performance variable values all in SAS® readable format.

The intent of this paper is not to explain MXG® software, which is SAS® source code, but to display some user examples of OS/390 computer performance graphs and reports as WEB GIFS and HTML output utilizing SAS Version 8® and the MXG® Computer Performance Database as input.

The WEB GIFS displayed were produced on an OS/390 mainframe using the SAS/Graph® GIF driver. The WEB GIFS and HTML reports are updated daily on the mainframe intranet Webserver for viewing via a WEB browser.

INTRODUCTION

The IBM SMF/RMF performance data is processed by the SAS/MXG software into daily, weekly and monthly MXG® Computer Performance Databases.

These databases contain numerous SAS® datasets. Each SAS® dataset contains performance variables in a SAS® readable format. The advantage of using MXG® software is that it converts raw performance data from IBM and third party computer vendors into SAS® readable format for processing by SAS® jobs.

MXG® provides numerous examples of SAS® plots and charts, from utilizing SAS/BASE to using SAS/GRAPH®.

For WEB graphics a change to MXG code examples was used to set up a macro to write the daily graphics and HTML to a corresponding daily file path representing a different day of the week i.e., SUN, MON, TUE...

Preparation - Setup SAS Macro for Daily GIFS

The SMF/RMF data is dumped at midnight by a SMFDUMP job submitted by the mainframe automated operations software.

Since the SMF/RMF data is dumped at midnight a %PICK_YESTERDAY macro was written for use by the daily SAS® programs to write the daily GIFS and HTML reports to the appropriate SUN, MON, TUE... file path on the OS/390 Webserver.

Figure #1 Pick_Yesterday SAS Macro

```
%MACRO PICK_YESTERDAY (NAME=' ');
/* USE DATA _NULL_ TO DEFINE _DAY MACRO */
DATA _NULL_;
FORMAT WEBSERVER_PATH $CHAR50.;

TODAY=TODAY();
YESTERDAY=TODAY-1;
/* DAY TAKES YESTERDAY VALUE MON, TUE...*/
DAY= UPCASE( PUT(YESTERDAY, WEEKDATE3.));

/* concatenate day and name onto the path */
WEBSERVER_PATH='/sasv8/mxg/' || DAY || &NAME;
/* put path in quotes & delete blanks */
WEBSERVER_PATH = QUOTE((TRIM(WEBSERVER_PATH)));

/* _DAY MACRO POINTS TO PATH AND DAY */
FILE INSTREAM RECFM=FB LRECL=80 BLKSIZE=800;
PUT @1 'MACRO _DAY '
    @12 WEBSERVER_PATH
    @63 '%';
RUN;

%INCLUDE INSTREAM;
RUN;
%MEND;
```

The PICK_YESTERDAY macro is displayed in Figure #1. The above macro with SAS Version 8 ODS is used to write the HTML and GIFS to the Webserver path. Initially the GIFS were written to a OS/390 PDS using a simpler version of the _DAY macro that converted to the day i.e., MON, TUE.... Then a batch job would copy the GIFS and HTML to the Webserver.

The character variable "webserver_path" is defined with the web path and DAY is concatenated to the path for the complete path. The variable day takes on the yesterday's value of SUN, MON, TUE... depending on the day the job was executed.

Then by using the FILE, PUT and %INCLUDE statements a _DAY macro is defined which points to the path in the OS/390 Webserver which has yesterday's value i.e., MON, TUE,...as appropriate. This permits Monday's GIFS and HTML reports to be written to the Webserver path '/sasv8/mxg/MON' when the SAS GIFS and report jobs are run on Tuesday morning just after midnight. Note the path is enclosed in quotes.

Previously with SAS 6.09E the SAS/GRAPH® GOPTIONS statement for the GSFNAME option was used and the _DAY macro only defined the day i.e., MON, TUE,... Then GSFNAME=_DAY ensured the GIFS were written to the MON, TUE... dataset defined in the execution JCL.

SETUP JCL - FOR DAILY GIFS

For SAS 6.09E the MON, TUE... file that the GIFS were written to was a partitioned dataset with a record length of 132 and a large or half track blocking on the OS/390 mainframe.

Figure #2 JCL Used with SAS6.09E

```

//*DOC:DAILY CHART#1- GIF DAILY CPU 24 HR PLOT
//*****
//GDCPU24P EXEC SAS609E,
//  OPTIONS='SOURCE NOMEMRPT MACRO',
//  WORK='12000,4000'
//INSTREAM DD UNIT=SYSDA,SPACE=(TRK,(1,1)),DISP=(,PASS
//FT20F001 DD UNIT=SYSDA,SPACE=(TRK,(1,1)),DISP=(,PASS
//LIBRARY DD DSN=MXG.FORMATS,DISP=SHR
//SOURCLIB DD DSN=MXG.USER.SOURCLIB,DISP=SHR
//
//PDB DD DSN=MXG.SOURCLIB,DISP=SHR
//PDB DD DSN=MXG.PDB.DAILY(0),DISP=SHR
//PDBCICS DD DSN=MXG.PDB.CICSTRAN(0),DISP=SHR
//SUN DD DSN=MXG.SUN.GIF(GDCPU24P),DISP=SHR
//MON DD DSN=MXG.MON.GIF(GDCPU24P),DISP=SHR
//TUE DD DSN=MXG.TUE.GIF(GDCPU24P),DISP=SHR
//WED DD DSN=MXG.WED.GIF(GDCPU24P),DISP=SHR
//THU DD DSN=MXG.THU.GIF(GDCPU24P),DISP=SHR
//FRI DD DSN=MXG.FRI.GIF(GDCPU24P),DISP=SHR
//SAT DD DSN=MXG.SAT.GIF(GDCPU24P),DISP=SHR
//SYSIN DD DSN=MXG.USER.SOURCLIB(GDCPU24P),DISP=SHR

```

The above JCL was one step of a multi-step job and repeated with a different sysin member. This example step would output the GIF daily CPU 24 hour plot to the appropriate MON, TUE, or WED... file.

However, by modifying the SAS programs to use ODS and the pick_yesterday macro with the Webserver path, the JCL no longer references the output dataset. This is still one step of a multi-step job to produce the daily OS/390 Web GIFS.

Figure #3 JCL Used with SASV8

```

//*****
//*DOC:DAILY CHART#1- GIF DAILY CPU 24 HR PLOT
//*****
//GDCPU24P EXEC SASV8,
//  OPTIONS='SOURCE NOMEMRPT MACRO',
//  WORK='12000,4000'
//INSTREAM DD UNIT=SYSDA,SPACE=(TRK,(1,1)),DISP=(,PASS
//FT20F001 DD UNIT=SYSDA,SPACE=(TRK,(1,1)),DISP=(,PASS
//LIBRARY DD DSN=MXG.FORMATS,DISP=SHR
//SOURCLIB DD DSN=MXG.USER.SOURCLIB,DISP=SHR
//
//PDB DD DSN=MXG.SOURCLIB,DISP=SHR
//PDB DD DSN=MXG.PDB.DAILY(0),DISP=SHR
//PDBCICS DD DSN=MXG.PDB.CICSTRAN(0),DISP=SHR
//SYSIN DD DSN=MXG.USER.SOURCLIB(GDCPU24P),DISP=SHR

```

The SAS/GRAPH® charts and plots produced for this presentation, utilize some of the basic MXG® DOCGRAF and GRAFXXXX examples as starting source code, but are then edited, tailored and modified to produce WEB GIFS and Reports of interest to our site and kept to a minimum of code for this presentation.

The SAS/Graph® charts and plots WEB GIFS charts and plots and HTML that are displayed used the MXG® PDB as input and the SAS Version 8 Output Delivery System ODS to create output.


Numerous charts and plots are updated daily to chart daily CPU usage for all shifts, tape cartridge usage including input or output tapes from SMF type 21 records, DASD percent full allocation from the SMF type19 records, DASD response time from type74 RMF records and CICS online response time from the type 110 SMF records.

In addition to daily charts, an additional group of charts for short term 15 day trends is updated daily to show immediate trending results that can result from various system or application changes.

The following is the OS/390 Web page page001.html for the main performance menu selection. This main performance menu has links to the SUN, MON, TUE... menu screens.

The menu also has links to the short term trend menu, which is similar to the daily menu's, and a link to download monthly charts in PowerPoint format.

Figure #4 Web Main Performance Menu



A daily menu screen follows. It is repeated for each day of the Week referencing a different day correspondingly with the SUN, MON, TUE files and charts. From this menu format twenty-eight daily charts and reports are easily reviewed. Also, the title GIF displays the date in the GIF.

Figure #5 Web Performance Menu for Monday



The SAS code to create the OS/390 date GIF is displayed in Figure #6. This source member GIFDATE utilizes PROC GPRINT to create the brilliant blue GIF with white lettering which contains the date in

Weekdate18 format and is displayed near the top of each perfsun, perfmon...perfsat menu web page.

This GIF is referenced by the static HTML code, but the GIF contains the updated date. Thus, when viewing the Monday Web menu page on Tuesday January 11, 2000. The GIF displays 'OS/390 Mon, Jan 10, 2000 Performance Charts - SAS/Graph and MXG.'

This GPRINT program doesn't use ODS but creates the GIF by using the PICK_YESTERDAY macro to specify a specific name '/GIFDATE.GIF' to append to the Webserver path.

Figure #6 SAS Code to Create Date GIF

```

/** GPRINT DAILY DATE AS A GIF FILE */
/* MEMBER=GIFDATE */

OPTIONS NODATE NONUMBER LABEL NOBYLINE;
/* SET LINE AND PAGE SIZE FOR PRINTOUT*/
OPTIONS LINESIZE=65 PAGESIZE=15;

/* create_day macro with a path & gif name*/
%INCLUDE SOURCLIB(V8MACROS); /*macros*/
%PICK_YESTERDAY(NAME='/GIFDATE.GIF')

FILENAME GIFOUT _DAY ;
/* Define where GIFDATE.GIF GOES */

/* SENDS PRINT OUTPUT TO DD CARD FT20F001 */
/* PRINT OUTPUT IS THE INPUT TO PROC GPRINT*/

PROC PRINTTO UNIT=20 NEW;
RUN;

DATA NULL ;
FORMAT TODAY WEEKDATE18.;
FILE PRINT NOTITLES;
TODAY=TODAY();
YESTERDAY=TODAY-1;
PUT @1 ' OS/390 -'
    @10 YESTERDAY WEEKDATE18.
    @29 'PERFORMANCE CHARTS - SAS/GRAPH & MXG';
RUN;


/* CLOSE THE PRINT FILE*/
PROC PRINTTO;
RUN;

GOPTIONS GUNIT=PCT RESET=GLOBAL DEVICE=GIF
HSIZE=5.6IN VSIZE=0.35IN
CBACK=BIB /*brilliant blue background */
HTEXT=45 FTEXT=SIMPLEX CTEXT=WHITE
GSFNAME=GIFOUT GSFMODE=REPLACE ;
RUN;

/* take the printout and output as a gif */
PROC GPRINT FILEREF=FT20F001 CTEXT=WHITE ;
RUN;

```

Figure #7 PLOT CPU Usage for 24 hours



SAS/GRAPH GPLOT with POINTLABELS

Initial Charts & Plots

The above plot of CPU usage in Figure #7 was produced using the SAS code in Figure#8. This is a plot of CPU utilization for the computer platform that includes the computer system 'PHLA'. The system identifier and date of the chart are displayed by using the #BYVAL1 and #BYVAL2 features in the title.

SAS Version 8 PointLabels are used and defined on the symbol statement, and are colored blue and positioned above each plot point.

The ODS statement creates a body labeled GDCPU24P.HTML in the Webserver Path /sasv8/mxg/MON/GDCPU24P.HTML that contains HTML code that references the plot at /sasv8/mxg/MON/gdcpu24p.gif. The plot name is defined in the plot statement using NAME='GDCPU24P'. The HTML code produced can be used to display the plot or a static web page which references the GIF can be used

MXG® creates a ZDATE variable which is the date the MXG® database was built. By using the DATE and ZDATE variables as a filter, the chart only has yesterday's value when ZDATE-DATE=1. This also acts as a filter in the event of multiple day's SMF/RMF data is input to the daily MXG PDB.

Since, this is GIF is a daily plot of CPU usage, it is named GD for GIF DAILY, then CPU24 for 24 hour of CPU displayed and then finally 'P' for plot - result GDCPU24P.

Figure #8 Procedure GPLOT - Plot CPU Usage.

```

/** CHART DAILY CPU USAGE - BY HOUR - 24HR PLOT */
/** JOB RUNS DAILY AND PUTS CHART INTO A GIF FILE **/

/* MEMBER=GDCPU24P DETAILED CPU USAGE BY PROCESSOR */
OPTIONS NODATE NONUMBER LABEL NOBYLINE;

%INCLUDE SOURCLIB(V8MACROS); /*CONTAINS MACRO DEF*/
%PICK_YESTERDAY
FILENAME ODSOUT _DAY ; /*PATH TO WRITE GIFS */

DATA CPUINFO;
SET PDB.RMFINTRV;
IF SYSTEM NE 'PHLA' THEN DELETE; /* LOOK AT PHLA */
TIME = TIMEPART(STARTTIME);
DATE = DATEPART(STARTTIME);
PCT=FLOOR(PLATBUSY); /* for platform and for beta V8 */
IF DATE NE ZDATE -1 THEN DELETE; /* only 1 day */
RUN;

ODS LISTING CLOSE;
ODS HTML PATH=ODSOUT BODY='GDCPU24P.HTML';

GOPTIONS RESET=GLOBAL DEVICE=GIF
GUNIT=PCT HSIZE=6.4IN VSIZE=4.5IN
FTEXT=SIMPLEX CTEXT=BLUE HTEXT=1.8
CBACK=YELLOW GSFMODE=REPLACE ;

PROC GPLOT DATA=CPUINFO; BY SYSTEM DATE;
FORMAT DATE WEEKDATE29.;
FORMAT TIME TIME5.0;
FORMAT PCTCUBY 3.0;
LABEL TIME='TIME*OF*DAY';

TITLE1 H=3.9 C=BLACK '#BYVAL1 COMPUTER SYSTEM';
TITLE2 H=3.8 C=RED 'DETAILED CPU ACTIVITY';
TITLE3 H=3.8 C=BLUE '#BYVAL2';

AXIS1 WIDTH=2 COLOR=BLUE /*VERTICAL CPU AXIS*/
LABEL=(H=2.8 COLOR=RED ROTATE=90 ANGLE=-90
'PERCENT*CPU')
MAJOR=(HEIGHT=2 COLOR=GREEN W=2)
MINOR=NONE
ORDER=(0 TO 100 BY 10)
VALUE=(H=2.8 COLOR=BLUE);

AXIS2 WIDTH=2 C=BLACK /*HORIZONTAL TIME AXIS*/
LABEL=(HEIGHT=2.9 COLOR=BLACK)
MAJOR=(HEIGHT=2 COLOR=RED)
MINOR=(H=1 N=5 C=BIB)
ORDER=(00:00'T TO '24:00'T BY '06:00'T)
VALUE=(H=2.9 COLOR=BLACK);

FOOTNOTE1 HEIGHT=1.8 C=BLACK
J=L 'SOURCE: MXG PERFORMANCE DATABASE'
J=R 'SAS/GRAPH PLOT';

FOOTNOTE2 HEIGHT=1 COLOR=RED
J=R 'GDCPU24P';

SYMBOL1 I=JOIN CI=RED L=1 W=2
V=SQUARE H=4 COLOR=GREEN
POINTLABEL=(COLOR=BLUE H=3.0 POSITION=TOP);

PLOT PCT*TIME /
HAXIS=AXIS2 VAXIS=AXIS1 CFRAME=WH
VREF=90 CVREF=GREEN
NAME='GDCPU24P';
RUN;

```

FIGURE #9 Procedure GCHART for 3D Chart

```

/* MEMBER=GDCPU24L Daily CPU USAGE BY LPAR */
OPTIONS NODATE NONUMBER NOBYLINE;
%INCLUDE SOURCLIB(V8MACROS);
%PICK_YESTERDAY
FILENAME ODSOUT _DAY;

DATA CPUINFO;
FORMAT LPAR $CHAR15.;
SET PDB.ASUM70PR;
IF SYSTEM NE 'PHLA' THEN DELETE;
TIME = TIMEPART(STARTTIME);
DATE = DATEPART(STARTTIME);
HOUR = HOUR(TIME);
/* FILTER TO AVOID 2 DAYS PERF DATA INTO CHART */
IF DATE NE ZDATE-1 THEN DELETE; /* YESTERDAY */
DUR=DURATM*NRPRCS; /* MULTIPLY DURATM BY # PROC*/
/* FIND CPU USAGE FOR THE 2 LPARS ON OUR SYSTEM */
LPAR=LP1NAME; PCT_CPU_USAGE=100*LP1UPDTM/DUR; OUTPUT;
LPAR=LP2NAME; PCT_CPU_USAGE=100*LP2UPDTM/DUR; OUTPUT;
LPAR=LPAR*OVERHEAD;
PCT_CPU_USAGE=100*LPPOPDTM/DUR; OUTPUT;
RUN;

ODS LISTING CLOSE;
ODS HTML PATH=ODSOUT BODY='GDCPU24L.HTML' ;

GOPTIONS GUNIT=PCT RESET=GLOBAL GSFMODE=REPLACE
DEVICE=GIF HSIZE=6.4IN VSIZE=4.5IN
CBACK=GOLD FTEXT=SIMPLEX CTEXT=RED HTEXT=1.8 ;
RUN;

PROC GCHART DATA=CPUINFO; BY DATE;
FORMAT PERCENT_CPU_USAGE 4.0;
FORMAT DATE WEEKDATE29.;

TITLE1 H=3.9 C=BLACK FONT=ZAPF 'OS/390 COMPUTER SYSTEM';
TITLE2 H=3.8 C=RED 'PERCENT CPU ACTIVITY BY HOUR';
TITLE3 H=3.8 C=BLUE '%BYVAL1';

AXIS1 WIDTH=2 COLOR=BLUE /*VERTICAL CPU AXIS*/
LABEL=(H=2.8 COLOR=RED ROTATE=90 ANGLE=-90
'PERCENT*CPU')
MAJOR=(HEIGHT=2 COLOR=GREEN W=2)
MINOR=NONE
ORDER=(0 TO 100 BY 10)
REFLABEL=(C=RED JUSTIFY=LEFT H=3.0
FONT=SIMPLEX TICK=10 "THRESHOLD=90%")
VALUE=(H=2.8 COLOR=RED);
AXIS2 WIDTH=2 C=BLACK /*HORIZONTAL TIME AXIS*/
LABEL=(HEIGHT=2.9 COLOR=BLACK)
ORDER=(0 TO 23 BY 1)
VALUE=(H=2.5 COLOR=GREEN);
FOOTNOTE1 HEIGHT=1.8 C=BLACK
J=L 'SOURCE: MXG PERFORMANCE DATABASE'
J=R 'SAS/GRAPH CHART';
FOOTNOTE2 HEIGHT=1 COLOR=RED
J=R 'GDCPU24L';

PATTERN1 COLOR=RED V=SOLID;
PATTERN2 COLOR=BLUE V=SOLID;
PATTERN3 COLOR=GOLD V=SOLID;

LEGEND1 VALUE=(HEIGHT=2.5 C=BLACK) ACROSS=3
CFRAME=WH CBORDER=RED CSHADOW=BLUE
LABEL=(H=2.5 C=MAGENTA POSITION=(BOTTOM CENTER));

VBAR3D HOUR / SUMVAR=PCT_CPU_USAGE
TYPE=SUM DISCRETE SUM SHAPE=PRISM
MAXIS=AXIS2 RAXIS=AXIS1 FRAME CFRAME=CXCCFF00
SUBGROUP=LPAR COUTLINE=ORANGE
LEGEND=LEGEND1 AUTOREF
NAME='GDCPU24L'; /*NAME FOR THE GIF CHART*/
RUN;

```

The SAS code in Figure #9 was used to create the vertical 3D bar chart in Figure # 10 with prism shaped bars. The prism shape is specified by the shape option in the VBAR3D statement. This is a hourly chart of CPU utilization for the Logical Partitions of a mainframe computer system. This information is in the MXG ASUM70PR dataset. The date of the chart is displayed by using the #BYVAL1 feature in the title.

The ODS statement creates a body labeled GDCPU24L.HTML in the Webserver Path /sasv8/mxg/THU/GDCPU24L.HTML that contains HTML code that references the chart gdcpu24l.gif. The chart name is defined in the VBAR3D statement using NAME='GDCPU24L'.

The legend statement uses the across=3 option to specify three entries across. This helps give some perspective to the production lpar, the test lpar and lpar overhead.

SAS variables can now be over eight characters in length and the PCT_CPU_USAGE calculated for the chart makes use of this feature.

The background color of the chart is specified in the options statement as cback=gold to select gold as the background color.

Autoref lines are also used and the SAS Version 8 'reflabel=' option with a "threshold=90" left justified is displayed on tick #10 to give some objective reference to the limits of the CPU chart.



Figure #10 Chart LPAR CPU Usage**SAS/GRAPH 3D Vertical Prism Bar Chart**

FIGURE #11 Tape Cartridge Activity



SAS/GRAPH 3D Vertical Block Bar Chart

The above chart in Figure #11 is a chart of 3480 tape cartridge usage on the OS/390 'PHLA' computer system. The OS/390 computer system name, the type of tape device type, and the date are displayed in the title by using #BYVALS.

The ODS statements direct the HTML output for the HTML body to GDTAPES.HTML.

The name of the chart is defined in the VBAR3D statement as name='GDTAPES' and is written as a GIF to the path /mxg/sasv8/TUE/gdtapes.gif.

The patterns for the 3D block chart are solid patterns and are specified on the pattern1 to pattern4 statements. Note that colors can be assigned using internet assignments of CX then a color such as 33FF33 in addition to such assignments as red, green or black.

The SMF record type21 contains tape information when a tape dismounts. MXG in the pdb.tapes dataset records the bytes read and written to the tape. This information is then used in this example to chart input tapes, output tapes, input/output tapes and tapes that were used but that no bytes read or written.

The block shape of the vertical bar charts is selected in the VBAR3D statement by using the option shape=block.

The chart displays that most of the tape usage is output tape during the midnight shift. This is expected with overnight backups.

FIGURE #12 GCHART Vertical 3D Chart

```

/* JOB RUNS DAILY - CHART INTO A GIF FILE */
/* MEMBER=GDTAPES */
OPTIONS NODATE NONUMBER LABEL NOBYLINE;
%INCLUDE SOURCLIB(V8MACROS);
%PICK_YESTERDAY
FILENAME ODSOUT _DAY;

DATA TAPE_INFO;
FORMAT TYPE $CHAR13.;
SET PDB.TAPES ;
IF SYSTEM NE 'PHLA' THEN DELETE; /* LOOK AT 1 SYS */
  HOUR = HOUR(TIMEPART(SMFTIME));
  DATE = DATEPART(SMFTIME);
  IF DEVICE NE '3480' THEN DELETE;
  IF ((BYTEWRIT GT 0) AND (BYTEREAD GT 0)) THEN
    TYPE='3-IN/OUT';
  ELSE IF BYTEWRIT GT 0 THEN TYPE='1-OUTPUT';
  ELSE IF BYTEREAD GT 0 THEN TYPE='2-INPUT';
  ELSE TYPE='4-NO DATA I/O';
  IF ZDATE-DATE NE 1 THEN DELETE;
  TAPE_COUNT=1;
RUN;
PROC SORT NODUP; BY SYSTEM DEVICE DATE; /*FOR BYVALS*/

ODS LISTING CLOSE;
ODS HTML PATH=ODSOUT BODY='GDTAPES.HTML';

GOPTIONS GUNIT=PCT RESET=GLOBAL DEVICE=GIF
CBACK=YELLOW HSIZE=6.4IN VSIZE=4.5IN
FTEXT=SIMPLEX CTEXT=BLACK HTEXT=2.0 GSFMODE=REPLACE ;

PROC GCHART DATA=TAPE_INFO; BY SYSTEM DEVICE DATE;
FORMAT DATE WEEKDATE29.;
LABEL TAPE_COUNT='TAPE*COUNT';
LABEL TYPE='TYPE*OF*USAGE';
LABEL HOUR='HOUR*OF*USAGE';

TITLE1 H=3.9 C=BLACK 'OS/390 #BYVAL1 COMPUTER SYSTEM';
TITLE2 H=3.8 C=RED 'TAPE #BYVAL2 CARTRIDGE ACTIVITY';
TITLE3 H=3.8 C=BLUE '#BYVAL3';

AXIS1 /*VERTICAL RESPONSE AXIS*/
  LABEL=(H=2.8 COLOR=RED ROTATE=90 ANGLE=-90)
  MAJOR=(HEIGHT=2 COLOR=GREEN W=2)
  MINOR=NONE
  VALUE=(H=2.8 COLOR=BLUE);
AXIS2 WIDTH=2 C=BLUE /*HORIZONTAL TIME AXIS*/
  LABEL=(HEIGHT=2.9 COLOR=BLACK)
  ORDER=(0 TO 23 BY 1)
  VALUE=(H=2.9 COLOR=MAGENTA);

FOOTNOTE1 HEIGHT=1.8
  J=L 'SOURCE: MXG PERFORMANCE DATABASE'
  J=R 'SAS/GRAPH CHART';
FOOTNOTE2 HEIGHT=1 COLOR=RED
  J=R 'GDTAPES';


PATTERN1 C=RED V=SOLID;
PATTERN2 C=BLACK V=SOLID;
PATTERN3 C=CX33FF33 V=SOLID;
PATTERN4 COLOR=CXFFA500 V=SOLID;

LEGEND1 VALUE=(HEIGHT=2.5 C=BLACK) ACROSS=4
  CFRAME=WH CBORDER=BLUE CSHADOW=GOLD
  LABEL=(H=2.5 C=MAGENTA POSITION=(BOTTOM CENTER));

VBAR3D HOUR / SUMVAR=TAPE_COUNT CFRAME=CX00FFFF
  DISCRETE RAXIS=AXIS1 MAXIS=AXIS2
  SUM LEGEND=LEGEND1 SUBGROUP=TYPE
  COUTLINE=YELLOW SHAPE=BLOCK
  NAME='GDTAPES';
RUN;

```

FIGURE #13 DASD I/O ACTIVITY



SAS/GRAPH 3D Horizontal Hexagon Bar Chart

The above chart in Figure #13 displays the OS/390 DASD I/O counts for the top 15 DASD on the 'PHLB' computer system for the date displayed. The SAS code for the GCHART SAS/GRAPH procedure is displayed in Figure #14.

The date displayed in the chart is taken from the input data. The OS/390 computer system and the date are displayed in the title by using the #BYVALS.

The option statement was used to limit the number of observations. By using proc means and sorting the data in descending order, only the top 15 DASD are selected by the option statement.

ODS is used to and the HTML code produced is in the GDVOLIOC.HTML file and referenced the CHART GIF that is named 'gdvolioc' for gif daily volume I/O chart.

The pattern1 statement is used to assign the cyan color to the bars on the horizontal bar chart.

The HBAR3D statement is used to create the horizontal bar chart with a 3D effect using HEXAGON shaped bars.

By specifying the sum option on the HBAR3D statement the DASD I/O counts for each volser are displayed to the right of each bar.

The sio74cnt for DASD I/O counts for each DASD volser is in the MXG TYPE74 dataset. This information is taken from the RMF type74 record on I/O activity.

FIGURE #14 GCHART Horizontal 3D Chart

```

/* MEMBER=GDVOLIOC DASD IO BY VOLSER */
OPTIONS NODATE NONUMBER LABEL NOBYLINE;
%INCLUDE SOURCLIB(V8MACROS);
%PICK_YESTERDAY
FILENAME ODSOUT _DAY ;

DATA DASDINFO;
SET PDB.TYPE74 ;
IF SYSTEM NE 'PHLB' THEN DELETE;
/* LOOK AT 1 SYSTEM*/
HOUR = HOUR(TIMEPART(STARTTIME));
DATE = DATEPART(STARTTIME);
IF ZDATE-DATE NE 1 THEN DELETE;
RUN;
PROC SORT NODUP DATA=DASDINFO;
BY SYSTEM DATE VOLSER;
PROC MEANS NOPRINT DATA=DASDINFO;
BY SYSTEM DATE VOLSER;
VAR SIO74CNT ; OUTPUT OUT=STATS SUM=SIO74CNT;

PROC SORT DATA=STATS; BY DESCENDING SIO74CNT;
RUN;
OPTIONS OBS=15; RUN; /*LIMIT TO 15 VOLSERS */

ODS LISTING CLOSE;
ODS HTML BODY='GDVOLIOC.HTML' PATH=ODSOUT;

GOPTIONS GUNIT=PCT RESET=GLOBAL DEVICE=GIF
HSIZE=6.4IN VSIZE=4.5IN GSFMODE=REPLACE
CBACK=CXFFF05F CTEXT=BLACK FTEXT=SIMPLEX
HTEXT=2.0;

PROC GCHART DATA=STATS ; BY SYSTEM DATE;
FORMAT DATE WEEKDATE29.;
FORMAT SIO74CNT COMMA18.0;
LABEL VOLSER='VOLSER';


TITLE1 H=3.9 C=BLACK
'OS/390 #BYVAL1 COMPUTER SYSTEM';
TITLE2 H=3.8 C=RED 'DASD VOLSER - I/O COUNTS';
TITLE3 H=3.8 C=BLUE '#BYVAL2';

AXIS1 /*VERTICAL AXIS*/
LABEL=(HEIGHT=2.6 COLOR=BLUE ROTATE=90
ANGLE=-90)
VALUE=(H=2.4 COLOR=BLACK);
AXIS2 WIDTH=2 COLOR=BLUE /*HORIZONTAL AXIS*/
LABEL=(H=2.8 COLOR=RED )
MAJOR=(HEIGHT=2 COLOR=RED W=2)
MINOR=NONE
VALUE=(H=2.6 COLOR=BLUE);
FOOTNOTE1 HEIGHT=1.8 COLOR=BLACK
J=L 'SOURCE: MXG PERFORMANCE DATABASE'
J=R 'SAS/GRAPH CHART';
FOOTNOTE2 HEIGHT=1 COLOR=RED J=R 'GDVOLIOC';

PATTERN1 COLOR=CYAN;
HBAR3D VOLSER/ TYPE=SUM SUMVAR=SIO74CNT SUM
DISCRETE RAXIS=AXIS2 MAXIS=AXIS1
DESCENDING CFRAME=CXFFA500
OUTLINE=BLACK SHAPE=HEXAGON
NAME='GDVOLIOC';
RUN;

```

FIGURE #15 PIE3D CHART I/O BY SHIFT



SAS/GRAPH PIE3D CHART

The above chart in Figure #15 displays the OS/390 DASD I/O counts by shift for system 'PHLA' for the date charted. The OS/390 computer system-id and the date are displayed in the title. The values are taken from the input data and displayed by using #byvals.

ODS is used and the HTML code produced is in the GDIOPIE.HTML file. The HTML code references the pie chart GIF that is named 'gdiopie' for the daily pie chart of I/O by shift.

The shift variable is created in the MXG RMFINTRV dataset, and contains a value for the shift that can be tailored by each computer site. Proc Format provided a more descriptive display of the shift, instead of using character values of 1,2,3 in the chart.

The pattern statements are used to describe the color of each pie slice. Pattern1 the first pattern (alphabetically selected i.e., day) for the day shift to pattern3 for the midnight shift. Patterns for the 3D charts are solid colors.

The PIE3D statement option of matchcolor is used to match the color of the text describing each pie slice with the color of each corresponding pie slice.

The DASD I/O counts are detailed in the MXG TYPE74 dataset and summarized in the RMFINTRV dataset.

FIGURE #16 PIE3D CHART I/O BY SHIFT

```

/** GIF DAILY IO PIE CHART */
/** MEMBER=GDIOPIE PIE CHART - IO BY SHIFT */
OPTIONS NODATE NONUMBER NOBYLINE;
%INCLUDE SOURCLIB(V8MACROS);
%PICK YESTERDAY
FILENAME ODSOUT _DAY;
PROC FORMAT;
    VALUE $SHIFTMT '1' = 'MIDNIGHT SHIFT'
                  '2' = 'DAY SHIFT'
                  '3' = 'EVENING SHIFT'
                  'W' = 'WEEKEND SHIFT';
RUN;

DATA DASD_IO_BYSHIFT;
FORMAT SHIFT $SHIFTMT.;
SET PDB.RMFINTRV;
IF SYSTEM NE 'PHLA' THEN DELETE;
IF ZDATE-DATE NE 1 THEN DELETE;
/*ONE DAY*/
RUN;

ODS LISTING CLOSE;
ODS HTML PATH=ODSOUT BODY='GDIOPIE.HTML' ;

GOPTIONS GUNIT=PCT RESET=GLOBAL
DEVICE=GIF HSIZE=6.4IN VSIZE=4.5IN
CBACK=CXE6E6FA FTEXT=ZAPF CTEXT=RED HTEXT=3.0
GSFMODE=REPLACE ;
RUN;

PROC GCHART DATA=DASD_IO_BYSHIFT;
    BY SYSTEM DATE;
FORMAT DATE WEEKDATE29.;

TITLE1 H=3.9 C=BLACK FONT=ZAPF
    'OS/390 #BYVAL1 COMPUTER SYSTEM';
TITLE2 H=3.8 C=RED
    'DASD I/O ACTIVITY BY SHIFT';
TITLE3 H=3.8 C=BLUE '#BYVAL2';

FOOTNOTE1 HEIGHT=1.8 C=BLACK
    J=L 'SOURCE: MXG PERFORMANCE DATABASE'
    J=R 'SAS/GRAPH CHART';

FOOTNOTE2 HEIGHT=1 COLOR=RED
    J=R 'GDIOPIE';

PATTERN1 COLOR=RED ;
PATTERN2 COLOR=CX0000FF;
PATTERN3 COLOR=BLACK ;

PIE3D SHIFT / SUMVAR=SIO74CNT
    OTHER=0 COUTLINE=YELLOW
    NOHEADING
    VALUE=NONE
    MATCHCOLOR
    SLICE=ARROW
    PCT=ARROW
    NAME='GDIOPIE'; /*NAME FOR THE CHART*/
RUN;

```


Figure #17 TABULATE Procedure

```

/**JOB RUNS DAILY AND PUTS table INTO A HTML FILE **/
/* MEMBER=DYCICSTB */
OPTIONS NODATE NONUMBER LABEL nobyline;

%INCLUDE SOURCLIB(V8MACROS);
%PICK_YESTERDAY
FILENAME ODSOUT _DAY;
RUN;
PROC FORMAT;
VALUE VAHOUR 0='MID TO 6:00 AM'
              6='06:00 - 07:00 AM'
              7='07:00 - 08:00 AM'
              8='08:00 - 09:00 AM'
              9='09:00 - 10:00 AM'
              10='10:00 - 11:00 AM'
              11='11:00 - 12 NOON'
              12='NOON- 1:00 PM'
              13=' 1:00 - 2:00 PM'
              14=' 2:00 - 3:00 PM'
              15=' 3:00 - 4:00 PM'
              16=' 4:00 - 5:00 PM'
              17=' 5:00 - 6:00 PM'
              18=' 6:00 - 7:00 PM'
              19=' 7:00 - MID';

RUN;
DATA RAWDATA;
FORMAT HOUR VAHOUR.;
FORMAT DATE WEEKDATE29.;
LABEL IRESPTM='INTERNAL*RESPONSE*TIME';
LABEL HOUR='HOUR*OF*INTERVAL';
SET
  PDBCICS.CICSTRAN;
DATE=DATEPART(STRTIME);
HOUR=HOUR(TIMEPART(STRTIME));
IF HOUR LE 5 THEN HOUR=0;
IF HOUR GE 19 THEN HOUR=19;
IF APPLID NE 'CICS4PRD' THEN DELETE;
IF (TRANNAME NE 'SUGI') THEN DELETE;
IF LENGTH(USER) ne 7 then delete;
_ZDATEM1

PROC SORT DATA=RAWDATA; BY DATE TRANNAME;
RUN;
ODS LISTING CLOSE; /* DO not produce a listing */
/* Prepare to produce a dataset */
ODS HTML PATH=ODSOUT BODY='DYCICSTB.HTML';

PROC TABULATE DATA=RAWDATA
  STYLE={JUST=right FOREGROUND=BLUE
  BACKGROUND=WHITE FONT_SIZE_2 FONT_WEIGHT=BOLD};

TITLE 'CICS ONLINE USER RESPONSE TIME';

LABEL TRANNAME='CICS*TRANSACTION*NAME';
CLASS DATE TRANNAME HOUR
  / STYLE={Background=blue FOREGROUND=WHITE
  FONT_SIZE=2 font_weight=BOLD};
CLASSLEV TRANNAME HOUR
  / STYLE={Background=white FOREGROUND=BLACK
  FONT_SIZE=3 font_weight=BOLD};
VAR IRESPTM / STYLE={Background=vpab /*pale blue */
  foreground=red FONT_SIZE=2
  font_weight=BOLD };

TABLE DATE*F=MMSS15.2, HOUR ALL='DAILY SUMMARY',
  TRANNAME*IRESPTM*(N*F=COMMA13.0 MEAN MIN MAX)
  /
  STYLE={BACKGROUND=green cellspacing=5
  bordercolor=yellow borderwidth=15 }
  BOX={label='CICS RESPONSE TABLE'
  style={background=white FONT_SIZE=3
  foreground=magenta}};

KEYWORD all /
  STYLE={BACKGROUND=cyan foreground=white };
KEYWORD N /
  STYLE={font_size=3 BACKGROUND=yellow };
keyword mean min max / s={font_size=3
  foreground=black background=pink };
KEYLABEL N='TRANSACTION COUNT';
KEYLABEL MEAN='AVG RESP MIN:SEC.TT';
KEYLABEL MIN='MIN RESP MIN:SEC.TT';
KEYLABEL MAX='MAX RESP MIN:SEC.TT';

```

```

RUN;
ODS HTML CLOSE;

```

The Tabulate procedure in Figure #17 produced the table in Figure #18 below. This output is not a SAS/GRAPH GIF file but is SAS/BASE HTML output. The output displays from a WEB browser.

CICS online transaction response data is available in the MXG CICSTRAN dataset. The online response and transaction data is displayed as a table instead of a SAS/GRAPH chart. This provides a tabular format of exact hourly transaction counts that some people prefer in addition to charts.

The four character CICS transaction name was changed for security reasons to 'SUGI', by editing the code and result. However, the result is the same for users seeking a beginning view of what can be done with SAS Version 8 Tabulate HTML output.

Figure # 18 Online Response Time
CICS ONLINE USER RESPONSE TIME

DATE Tuesday, August 3, 1999				
CICS RESPONSE TABLE	CICS*TRANSACTION*NAME			
	SUGI			
	INTERNAL*RESPONSE*TIME			
	TRANSACTION COUNT	AVG RESP MIN:SEC.TT	MIN RESP MIN:SEC.TT	MAX RESP MIN:SEC.TT
HOUR*OF*INTERVAL				
MID TO 6:00 AM	1,017	0:00.02	0:00.01	0:02.61
06:00 - 07:00 AM	801	0:00.09	0:00.01	0:01.13
07:00 - 08:00 AM	2,159	0:00.10	0:00.00	0:02.74
08:00 - 09:00 AM	4,962	0:00.12	0:00.00	0:02.09
09:00 - 10:00 AM	6,927	0:00.14	0:00.00	0:03.10
10:00 - 11:00 AM	7,337	0:00.14	0:00.00	0:08.69
11:00 - 12 NOON	7,722	0:00.11	0:00.00	0:01.46
NOON- 1:00 PM	5,190	0:00.09	0:00.00	0:01.94
1:00 - 2:00 PM	7,789	0:00.12	0:00.00	0:02.07
2:00 - 3:00 PM	7,610	0:00.13	0:00.01	0:13.45
3:00 - 4:00 PM	6,403	0:00.13	0:00.00	0:02.94
4:00 - 5:00 PM	4,064	0:00.09	0:00.00	0:02.13
5:00 - 6:00 PM	2,236	0:00.08	0:00.00	0:01.12
6:00 - 7:00 PM	366	0:00.07	0:00.01	0:01.07
7:00 - MID	1,112	0:00.03	0:00.01	0:01.80
DAILY SUMMARY	65,695	0:00.12	0:00.00	0:13.45

Procedure Tabulate - HTML Output

CONCLUSION

Utilizing BASE SAS and SAS/GRAPH® plus the OS/390 Webserver to display the updated daily WEB GIFS and HTML reports from the MXG® performance database improved the ease of use for managers and technicians to review daily performance activity.

The programs provides quality graphical displays of computer performance data. The results can be easily interpreted by tactical performance technicians for investigating “bottlenecks” or by management to review for overall system performance and trends.

Providing BASE SAS HTML reports from the Print, Report or Tabulate procedures also provide additional information via the intranet. The productivity of the staff increased by eliminating searches for lost or miss-routed printed reports.

The Web browser intranet format of providing information is also appealing to many staff members and management. The display of graphs and HTML reports on the intranet web server improved ease of use in reviewing performance data and increased the use of MXG® and SAS/GRAPH® software. Overall productivity of the staff also increased by making the charts and reports more easily available for review by the staff and managers needing access to the information.

REFERENCES

IBM OS/390 - MVS Initialization and Tuning Reference - SC28-1752-02.

IBM OS/390 - MVS System Management Facilities (SMF) - GC28-1783-02

IBM OS/390 - Resource Measurement Facility (RMF) User's Guide - SC28-1949-01.

Merrill's Expanded Guide to Computer Performance Evaluation Using the SAS System.

Merrill's Expanded Guide Supplement.

Technical Newsletters for Users of MXG®.

MXG Archives at WWW.MXG.COM

Computer Measurement Information at www.cmg.org.

Animated GIFS from well known computer measurement expert Dr. Bernie Domanski at <http://domanski.cs.csi.cuny.edu/index.html>.

NESUG Proceeding 1994. Effective Poster Presentations for SAS® User Group Conferences: Helen-Jean Talbott

SAS® User's Guide: Basics.

SAS/GRAPH® User's Guide.

What's New in SAS® Software for Version 7 and the Version 8 Developer's Release.

SAS Web Site at WWW.SAS.COM

SAS ONLINE DOC Version 7 CDROM.

SAS ONLINE DOC Version 8 CDROM.

Author Contact

Neal Musitano Jr.
Department of Veterans Affairs
Benefits Delivery Center-284/31
P. O. BOX 7545
Philadelphia, PA. 19101-7545

Phone 215-842-2000 ext. 4102
FAX 215-381-3456
EMAIL TSDNMUSI@VBA.VA.GOV